


QB64 – Interprete de BASIC moderno para sistemas de 32 y 64 bits

Publicado el [El Informatico](#) - 14 de abril de 2022 -



```

MANDELT.BAS - QB64 x64
File Edit View Search Run Debug Options Tools Help
MANDELT.BAS
22 maxIteration = 223
23 While (x * x + y * y <= (2 * 2) * (16384 * z) And iteration < maxIteration)
24   xtemp = x * x - y * y + x0
25   y = (2 * x * y + y0) / z + (z * MANY)
26   x = xtemp / z + (z * MANX)
27   iteration = iteration + 1
28 Wend
29 If iteration <> maxIteration Then
30   c = iteration
31 Else
32   c = 0
33 End If
34 PSet (x0, y0), c + 32
35 Next
36 Next
37 Locate 1, 1: Print "Z: " + Str$(z) + "X:" + Str$(MANX) + " Y:" + Str$(MANY) + " Pc:" + Str$(MANMUL)
38 If z > 96 Then
39   z = z + STEPZ
40   STEPZ = STEPZ + 0.01
41 End If
42 _Delay 0.1
43 If InKey$ = Chr$(27) Then
44   Exit For
45 End If
46 If UCase$(InKey$) = "D" Then
47   MANX = MANX + MANMUL

```

QB64

BASIC es un lenguaje muy antiguo. Fué muy popular en los años 80, donde casi todos los ordenadores de la época integraban su propia implementación del lenguaje de programación, normalmente integrado en la propia BIOS del sistema. Pero si hay algo por lo que la gente se acuerda de éste lenguaje es porque Microsoft integró un interprete de QuickBASIC llamado QBASIC, en algunas de sus versiones más modernas de MS-DOS y en los sistemas de Windows 9x (incluido en el CD-ROM del sistema).


Sin embargo, con la llegada de los sistemas de 32 y 64 bits éste interprete ha quedado obsoleto. Sobre todo con otros lenguajes como Python, Java o similares. Pero si aún quieres escribir o incluso ejecutar programas para éste intérprete, como en los viejos tiempos, existen interpretes más modernos que son compatibles con los programas antiguos. QB64 es uno de ellos.

QB64

QB64 es un intérprete moderno para el lenguaje QuickBASIC de Microsoft. Incluye soporte para la sintaxis y las instrucciones antiguas que incorporaba QBasic, y añade además algunas instrucciones nuevas con funciones más modernas.

Por ejemplo, si bien incluye soporte para gráficos estilo VGA al igual que lo hacía QBASIC, también incluye nuevas funciones que nos permiten manipular colores e imágenes, incluso con canal de transparencia.

También nos añade funciones para controlar el ratón, usar fuentes, usar un joystick o mando, interactuar con la memoria, interactuar con el sistema operativo, sonidos, e incluso soporte para TCP/IP. Todas éstas funciones sólo podrían estar disponibles a través de librerías externas en QBasic, pero QB64 las incorpora. Puedes ver una lista de funciones en la [wiki](#).



GORILLAS.BAS en QB64

QB64 soporta todas las instrucciones de QBasic – o casi. Algunas funciones más avanzadas, como DEF FN, aún no están implementadas (DEF SEG sí lo está). Además al ser un emulador, no funciona de manera 100% exacta a como lo haría QBasic, por lo que es posible que haya diferencias en como se ejecuta el código.

No obstante, puedes reemplazar DEF FN con una simple función:

```
DEF FNran (x) = Int(Rnd(1) * x) + 1:  
  
Function ran (x)  
    ran = Int(Rnd(1) * x) + 1  
End Function
```

Cómo obtenerlo

Puedes descargar QB64 desde la página oficial [QB64.org](https://qb64.org). Es gratuito y de código

libre, y curiosamente está escrito en el mismo lenguaje, con lo que es posible ejecutarlo y compilarlo desde el propio editor de QB64.


Una vez descargado, abre el comprimido y descomprime la carpeta qb64 donde quieras tener el programa. El ejecutable que hay dentro (qb64.exe) contiene el interprete.


¿Cómo se usa?

Similar a QBasic o Microsoft QuickBASIC, pero con funcionalidades de IDE modernas.

Por ejemplo, además de incorporar coloreado de sintaxis, disponemos de la posibilidad de usar puntos de interrupción para depurar el código, inspeccionar valores de variables, e incluso la posibilidad de compilar el código en un .exe. También podemos seleccionar código y usar los atajos del teclado para cortar, copiar o pegar selecciones (algo que no es posible en QBasic).


Símplemente introduce el código, guarda el archivo (File > Save) y después ejecutalo. QB64 generará un ejecutable por cada programa automáticamente, de modo que también es posible distribuir nuestros programas y juegos como ejecutables para Windows y Linux (dependiendo de la versión que estemos usando).





Ejemplos

MANDELBT.BAS – Genera un fractal Mandelbrot y hace zoom hacia dentro.



MANDELBT.BAS

```

Screen 13
Window (-2, 1.5)-(-2, -1.5)

Print "Pulsa S para comenzar..."
While UCase$(InKey$) <> "S"
    Sleep 1
Wend

MAXZ% = 1024
STEPZ = .2


MANX = 0.4427043
MANY = 0.3782007
MANMUL = 0.005

For z = 0.1 To MAXZ% Step STEPZ
    For x0 = -2 To 2 Step .01
        For y0 = -1.5 To 1.35 Step .01
            x = 0
            y = 0
            iteration = 0
            maxIteration = 223
            While (x * x + y * y <= (2 * 2) * (16384 * z) And iteration <
maxIteration)
                xtemp = x * x - y * y + x0
                y = (2 * x * y + y0) / z + (z * MANY)
                x = xtemp / z + (z * MANX)
                iteration = iteration + 1
            Wend
            If iteration <> maxIteration Then
                c = iteration
            Else
                c = 0
            End If
            PSet (x0, y0), c + 32
        Next
    Next
    Next
    Locate 1, 1: Print "Z: " + Str$(z) + "X: " + Str$(MANX) + " Y: " +
Str$(MANY) + " Fc: " + Str$(MANMUL)
    If z > 96 Then
        z = z + STEPZ
        STEPZ = STEPZ + 0.01
    End If
    _Delay 0.1
    If InKey$ = Chr$(27) Then
        Exit For
    End If
    _Delay 0.01
    MANMUL = MANMUL / 1.002
Next

```

QWORDLE.BAS – Wordle en QB64 (en Inglés). Para funcionar correctamente, requiere que descargues el archivo [words.txt](#) en el mismo directorio donde tengas el programa. El juego funciona exáctamente como el Wordle original, excepto que la palabra se genera al azar al empezar una partida nueva en lugar de ser una palabra diaria.

Tienes 6 intentos para adivinar la palabra de 5 letras. En cada intento, debes introducir una palabra de 5 letras que exista dentro del diccionario. Por cada letra de la palabra introducida, se mostraran en gris las letras que no se incluyan en la palabra a adivinar. En naranja las que sí que se incluyan pero que se encuentren en la posición errónea, y en verde las letras correctas.



QWORDLE.BAS

```

Const LINEHEIGHT% = 50
Const BOXSIZE% = 32
Const GUESS% = 2
Const WRONG% = 6
Const NOTIN% = 8

Randomize Timer

Open "words.txt" For Input As #1
line$ = "-"
linen% = 0
Dim WordArray$(2315)
Do Until EOF(1)
    Line Input #1, line$
    WordArray$(linen%) = line$
    linen% = linen% + 1
Loop
Close #1

Intro
start:
Locate 1, 1
Screen 12
Cls
Line (0, 0)-(640, 480), 0, BF

word$ = UCase$(WordArray$((Rnd * 2315) + 1))

Locate 1, 1: Print "Enter a 5 letter word when prompted. Press ENTER to
proceed."

For try% = 1 To 6 Step 1

```

```

for try% = 1 to 6 Step 1
    init:
    cl 2
    Locate 2, 1: Input "> ", inpt$

    If inpt$ = "" Then
        cl 2
        Locate 2, 1: Print "You must enter a 5 letter word. Press ENTER
to try again."
        Do
            Loop Until InKey$ = Chr$(13)
            GoTo init
        End If
    If Len(inpt$) <> 5 Then
        cl 2
        Locate 2, 1: Print "You entered a " + Str$(Len(inpt$)) + " letter
word. It must be a 5 letter word."
        Do
            Loop Until InKey$ = Chr$(13)
            GoTo init
        End If

    For n = 0 To 2315 Step 1
        floop:
        If UCase$(WordArray$(n)) = UCase$(inpt$) Then
            PrintWord UCase$(inpt$), UCase$(word$), try%
            If UCase$(inpt$) = UCase$(word$) Then
                Locate 26, 1: Print "Congratulations! You guessed the
word in " + Str$(try%) + " attempts! Wanna try again?"
                Locate 27, 1: Print "Press R to retry, or E to exit"
                Do
                    usi$ = UCase$(InKey$)
                    If usi$ = "R" Then
                        GoTo start
                    End If
                    If usi$ = "E" Then
                        Stop
                    End If
                Loop Until UCase$(InKey$) = "R" Or UCase$(InKey$) = "E"
                End If
                try% = try% + 1
                If try% > 6 Then
                    GoTo gameover
                End If
                GoTo init
            End If
        Next
        cl 2
        Locate 2, 1: Print "The word " + UCase$(inpt$) + " is not in the
dictionary. Try again!"
        Do
            Loop Until InKey$ = Chr$(13)
            GoTo init
        Next
    gameover:
    Locate 26, 1: Print "Tough luck! The correct answer was " + word$
    Do
        usi$ = UCase$(InKey$)
        If usi$ = "R" Then
            GoTo start
        End If
        If usi$ = "E" Then
            Stop

```

```

    End If
Loop Until UCase$(InKey$) = "R" Or UCase$(InKey$) = "E"

Sub cl (l)
    For i = 1 To 80
        Locate l, i: Print " "
    Next
End Sub

Sub PrintWord (guess$, word$, try%)
    row = 2 + 3 * try%
    For i = 1 To 5 Step 1
        colr% = NOTIN%
        charguess$ = Mid$(guess$, i, 1)
        charsol$ = Mid$(word$, i, 1)
        col = i * 6 + 1

        If charguess$ = charsol$ Then
            colr% = GUESS%
        Else
            For x = 1 To 5 Step 1
                If charguess$ = Mid$(word$, x, 1) Then
                    colr% = WRONG%
                End If
            Next
        End If

        Locate row, col
        DrawSquare col * 8 - 18, row * 16 - 18, 32, colr%
        biglt charguess$, 15, colr%
    Next
End Sub

Sub DrawSquare (x%, y%, d%, scolor%)
    Line (x%, y%)-(x% + d%, y% + d%), scolor%, BF
End Sub

Sub biglt (text$, fore%, back%)
    Dim pixel(8, 16)
    For i = 1 To Len(text$)
        Print Mid$(text$, i, 1);
        xul = 8 * (Pos(0) - 2): yul = 16 * (CsrLin - 1)
        For y = 0 To 15
            For x = 0 To 7
                pixel(x, y) = Point(xul + x, yul + y)
            Next
        Next
        Line (xul, yul)-(xul + 15, yul + 28), back%, BF
        For y = 0 To 15
            For x = 0 To 7
                If pixel(x, y) Then
                    PSet (xul + 2 * x, yul + 2 * y), fore%
                    PSet (xul + 2 * x + 1, yul + 2 * y), fore%
                    PSet (xul + 2 * x, yul + 2 * y + 1), fore%
                    PSet (xul + 2 * x + 1, yul + 2 * y + 1), fore%
                End If
            Next
        Next
        Locate , Pos(0) + 1
    Next
End Sub

```



```
Sub Intro ()
  Screen 12
  Locate 2, 26: Print "QWORLD - BY AARON C.D.C
  (https://elinformati.co)"
  Locate 3, 26: Print "-----"
  Locate 10, 1: Print "Enter a 5 letter word when prompted, and then
press ENTER."
  Locate 11, 1: Print "You have 6 attempts to guess the correct word."
  Color GUESS%: Locate 12, 6: Print "When a letter is in the right
place, it will be displayed in green."
  Color WRONG%: Locate 13, 6: Print "When a letter is in the wrong
place but exists in the word, it will"
  Color WRONG%: Locate 14, 6: Print "be displayed in orange."
  Color NOTIN%: Locate 15, 6: Print "When a letter is not part of the
word, it will be displayed in gray."
  Color 15: Locate 16, 1: Print "You can only input words that exist in
the ENGLISH dictionary."
  Locate 20, 1: Print "Press ENTER to Start"
  Do
  Loop Until InKey$ = Chr$(13)

End Sub
```



ANTERIOR

[Juegos y aplicaciones de MS-DOS con DOSBOX \(Actualizado\)](#)

SIGUIENTE

[Extraer textos e imágenes de documentos PDF y otras imágenes con Python, PyMuPDF y OCRMyPDF](#)

Buscar ...



Entradas Recientes

- [Estoy hasta las narices de la web moderna](#)
- [Ingeniería inversa básica con Ghidra](#)
- [Acerca de la nueva ley transgénero \(Y sobre la disforia de género\)](#)
- [Depresiones causadas por las redes sociales](#)
- [¿Necesito saber matemáticas para aprender informática?](#)
- [¿Es el fin de los discos duros tradicionales?](#)

Categorías

[Actualidad](#)[Android](#)[Básicos](#)[Ciberseguridad](#)[Clima](#)[Criptografía](#)[Electronica](#)[Emulación / Virtualización](#)[FOSS](#)[Hacking](#)[Hardware](#)[Informática](#)[Internet](#)[Juegos](#)[Opinion](#)[Otros](#)[Personal](#)[Privacidad](#)[Programación](#)[Tecnología](#)[Time Machine](#)[Tutoriales](#)

RSS

[Subscribirse al feed RSS](#)

Inicio
Catálogo
PDFs
Manuales
Política de privacidad
Política de Cookies
Acerca de mi
Acerca de ElInformati.co